
Nummernschalter-Prüfsoftware R2.0
Die nachfolgende Dokumentation beschreibt die Installation sowie die Anwendung einer Excel-basierten Prüfsoftware für Nummernschalter. Der dazu gehörige Quellcode ist frei verfügbar und aus Gründen der Transparenz und der Erweiterbarkeit für eigene Zwecke erläutert.
Die Erläuterungen beziehen sich auf das Release R2.1 der Software. Gegenüber Version 2.0 wurde im Teil „Bedienung“ der Teil „Zuordnung der Kabel des Nummernschalters“ hinzugefügt. Die in R2.0 fehlerhafte Zuordnung des Anschlusskabel im Kapitel „Hardware“ wurde korrigiert.

Der verwendete Code baut im Wesentlichen auf dem von Andreas Panskus unter http://www.wasser.de/telefon-alt/forum/index.pl?job=thema&tnr=100000000006241&seite=1&begriff=&tin=&kategorie= vorgestellten Programm sowie der Bibliothek RSAPI.DLL von H.-J. Berndt und B. Kainka auf.
Disclaimer:

Das Programm sowie die vorgestellten Prozeduren können frei verwendet werden, jedoch kann für eine einwandfreie Funktion in allen möglichen Situationen keine Gewähr übernommen werden. Ferner wird nicht für etwaige Schäden oder Verluste gehaftet, die durch die Verwendung dieser Software bzw. der in dieser Dokumentation dargelegten Informationen entstehen. Der Benutzer erklärt sich ausdrücklich damit einverstanden, das Programm sowie die in dieser Dokumentation dargelegten Informationen auf eigene Verantwortung zu verwenden.

Microsoft, Excel und Windows sind eingetragene Warenzeichen der Microsoft Corporation.

.

Installation

Eigentlich ist für die Nummernschalter-Prüfsoftware nicht viel zu installieren. Die Software setzt einen unter Windows laufenden Rechner mit Excel >= Version 2003 voraus, der über eine serielle Schnittstelle verfügt. In der Regel muss lediglich eine Installation der RSAPI.DLL erfolgen. Dazu ist diese zunächst von der entsprechenden Website downzuloaden (s. Schnittstelle Excel <-> RS 232: RSAPI.DLL) und dann durch Doppelclick zu installieren). Zum Aufruf der Prüfsoftware ist die Excel-Datei zu öffnen. Dabei ist zu gewährleisten, dass die Sicherheitseinstellungen des Systems nicht so hoch sind, dass Makros automatisch abgeschaltet werden. Über den Menuepunkt „Extras/Makro/Sicherheit“ in Excel kann die Sicherheitsstufe in diesen Fällen auf „Mittel“ heruntergesetzt werden.
Hardware

Um die Nummernschalter-Prüfsoftware starten zu können, ist ein unter Windows laufender Rechner mit Excel >= Version 2003 erforderlich, der über eine serielle Schnittstelle verfügt. Von dieser werden lediglich die Leitungen RTS, DSR und CTS verwendet. Dazu wird an eine typischerweise 9-polige D-Sub-Buchse an die Pins 6, 7 und 8 wie folgt ein vieradriges Kabel angelötet: Pin 6 br, Pin 7 ws und ge, Pin 8 gn. Die anderen Seiten der Litzen werden mit Krokodilklemmen versehen. Anschließend werden die Klemmen wie folgt an den Probanden angesteckt:

Nummernschalter:

ge und gn: nsi/nsr

ws und br: nsa

Telefon:

ge und gn: La/Lb
Über die RTS-Leitung erhält der Nummernschalter einen log. 1 Pegel, der bei geschlossenem nsi/nsr-Kontakt auf dem CTS-Eingang bzw. geschlossenem nsa-Kontakt auf dem DSR-Eingang detektiert wird.

Die Art der Leitung sollte im Bereich 1m unkritisch sein und keiner Schirmung bedürfen.
ACHTUNG: In der vorangegangenen Doku war die Zuordnung der Kabelfarben zu den Pins des D-Sub-Steckers fehlerhaft.
Bedienung

Zum Gebrauch der Prüfsoftware sind lediglich das Excel-File zu öffnen und das Testkabel in die serielle Schnittstelle einzustecken.
Der nsi/nsr-Kontakt des Nummernschalters (bzw. die a/b Kontakte eines Telefons) und ggf. der nsa-Kontakt sind nun wie folgt an das Testkabel anzuschließen:
Nummernschalter:

ge und gn: nsi/nsr

ws und br: nsa

Telefon:

ge und gn: La/Lb
Anschließend wird die Software im Blatt „Parameter“ parametriert. Bei den gelben Feldern handelt es sich um Felder, in die frei Werte eingegeben werden können, bei dem hellgrünen Feld handelt es sich um ein Auswahlfeld.
[image: image1.emf]Parameter Wert Kommentar

PRELLZEIT 0,02 Impulsdauer kleiner dieser Zeitangabe in Sekunden wird als Prellen gewertet

WARTEZEIT 3 Maximale Zeitdauer in Sekunden, die auf das Eintreffen von Impulsen gewartet wird

COMPORT COM1 Com-Port, über den die Messung erfolgt


Im Wert zur „PRELLZEIT“ wird die Zeitdauer in Sekunden angegeben, innerhalb derer aufeinanderfolgende Pegelwechsel des nsi/nsr-Kontaktes als Prellen gewertet werden (der nsa-Kontakt wird nicht auf Kontaktprellen hin untersucht).

Die „WARTEZEIT“ gibt an, wie viele Sekunden nach dem letzten Impuls das Ergebnis dargestellt wird bzw. wie lange man nach Drücken des „Start“ Buttons Zeit hat, einen Wählvorgang einzuleiten.

Mit „COMPORT“ (Auswahlfeld) wird bestimmt, über welchen seriellen Port die Messung erfolgt.

Da diese Einstellungen typischerweise beibehalten werden, empfiehlt es sich, nach Setzen der Parameter die Datei zu speichern. Anschließend wechselt man ins Blatt „Test“, von dem aus die Testroutine gestartet wird.

nsi/nsr

Zeit Pegel Dauer /

ms

Abw.

zu Soll

Ges.dauer

1 Impuls /ms

Abw.

zu Soll

Kennzeichnung:

0 7128,06 1

1 7129,62 0 70,6 13,9%

2 7129,69 1 39,9 5,0%

3 7129,73 0 63,8 2,9%

4 7129,79 1 39,3 3,4% Angeschlossenes Gerät:

5 7129,83 0 62,1 0,2%

6 7129,90 1 38,1 0,3% Ist Soll Min. Max.

7 7129,93 0 62,7 1,1% Gewählte Ziffer:

10 10

8 7130,00 1 40,1 5,5% Kontaktprellen:

5 15

9 7130,04 0 63,8 2,9% Mittelwert offen (3.-letzte Flanke) / ms: 62,9 62,0

10 7130,10 1 39,6 4,2% Mittelwert geschlossen (3.-letzte Flanke) / ms: 39,2 38,0

11 7130,14 0 61,9 -0,2% Maximalwert offen (3.-letzte Flanke) / ms:

63,9

12 7130,20 1 38,2 0,5% Minimalwert offen (3.-letzte Flanke) / ms:

61,9

13 7130,24 0 62,9 1,5% Schwankungsbreite offen vom Mittel:

3% 10%

14 7130,30 1 40,1 5,5% Maximalwert geschlossen (3.-letzte Flanke) / ms:

40,1

15 7130,34 0 63,9 3,1% Minimalwert geschlossen (3.-letzte Flanke) / ms:

38,1

16 7130,41 1 39,5 3,9% Schwankungsbreite geschlossen vom Mittel:

5% 10%

17 7130,45 0 62,1 0,2% Mittelwert Gesamtdauer 1 Impuls / ms:

102 100 90 110

18 7130,51 1 38,4 1,1% Dauer Wählvorgang / ms: 990 960 866 1058

19 7130,55 0 62,8 1,3% Offen/Geschlossen-Verhältnis (aus Flanke 3-19) :1 : 1,61 1,63 1,3 1,9

Status:

0 7128,06 0

1 7129,00 1 1.608,80

2 7130,61 0

100,20 0,2%

102,80 2,8%

110,50 10,5%

103,10 3,1%

103,40 3,4%

100,10 0,1%

100,50 0,5%

103,00 3,0%

103,40 3,4%

Im hellgrünen Feld ist anzugeben, um welchen Typ es sich bei dem angeschlossenen Gerät handelt. Hier kann zwischen einem Nummernschalter und einem Telefon unterschieden werden. Die Auswahl hat Einfluss auf die Spalten der obigen Ergebnistabelle, die angezeigt werden, sowie auf die Einbeziehung des nsa-Kontakts in die Prüfung: Im Falle eines Telefons werden die Spalten „Dauer / ms“ und „Abw. zu Soll“ ausgeblendet, da die Werte aufgrund des Einflusses des Funkenlöschkondensators nicht sinnvoll interpretierbar sind. Auch erfolgt keine Prüfung des nsa-Kontakts.
Zum Starten der Messung ist der Button „Start“ zu drücken und dann innerhalb der im Blatt Parameter durch den Wert von „WARTEZEIT“ definierten Zeitspanne die Ziffer 10 zu wählen. Ist der Nummernschalter defekt und werden keine Impulse detektiert, erfolgt eine entsprechende Fehlermeldung, andernfalls werden die Messwerte angezeigt. Indem das getestete Gerät im Feld „Kennzeichnung“ entsprechend bezeichnet wird, können die Messwerte für eine spätere Nutzung gespeichert werden. Dazu ist nach Eingabe der Kennzeichnung der Button „Save“ zu drücken.

Durch Drücken des Buttons „Load“ können gespeicherte Werte wieder zurückgeladen werden.

Zuordnung der Kabel des Nummernschalters

Bei Nummernschaltern des Typs 38 und jünger entsprechen die Farben der Anschlusskabel in der Regel denen des „Testadapters“, d.h. ge und gn: nsi/nsr, ws und br: nsa. Sollten die Farben nicht mehr erkennbar sein, so können bei einem grundsätzlich funktionierenden Nummernschalter die richtigen Leitungspaare folgendermaßen herausgefunden werden:

1. Nur das gelbe und das grüne Kabel des Testadapters an zwei beliebige Kabel des Nummernschalters anschließen

2. Die Software in den Prüfmodus „Telefon“ stellen.

3. Die Ziffer 0 wählen.

a. Werden mehrere Impulse erkannt, ist der nsi/nsr-Kontakt richtig angeschlossen. Dann die Kabel des nsa-Kontakts an die anderen beiden Strippen des Prüfadapters anschließen.

b. Wird nur ein Impuls erkannt, wurde der nsa- anstelle des nsi/nsr-Kontakts angeschlossen. In diesem Fall das gelbe und grüne Kabel des Testadapters an die anderen beiden Kabel des Nummernschalters anschließen, an die gerade geprüften Kontakte das weiße und braune Kabel des Testadapters anschließen.
c. Wurde gar kein Impuls erkannt, dann das gelbe Kabel des Testadapters an eines der beiden freien Kabel des Nummernschalters anschließen und Schritt 3 wiederholen.

4. Die Prüfsoftware in den Prüfmodus „Nummernschalter“ stellen. Nun kann die Prüfung des Nummernschalters erfolgen.

Tabellenblätter

Tabellenblatt „Test“
Das Tabellenblatt „Test“ stellt quasi das Regiezentrum dar. Von hier aus werden die Funktionalitäten des Programms aufgerufen.

Beim Test werden die Zeitdauern der einzelnen Impulse ermittelt und in der Tabelle A3:D26, A29:D31 und L10 abgelegt. Spalte A beschreibt die durchgehende Nummerierung der Flanken, Spalte B den jeweiligen Zeitstempel des Ereignisses (s.a. Hochgenaue Zeitmessung in Windows,
QueryPerformanceFrequency und QueryPerformanceCounter), Spalte C, ob mit der jeweiligen Flanke ein Wechsel auf log. 0 (=Unterbrechung) oder log. 1 (=Durchgang) erfolgt ist. Spalte D beschreibt die Dauer vom Beginn der betreffenden Flanke bis zum Beginn der nächsten Flanke. In Abhängigkeit davon, ob es sich um eine offen- oder eine geschlossen-Phase handelt, wird in Spalte E die prozentuale Abweichung zum Sollwert ermittelt. Spalte F gibt die Gesamtdauer für eine offen-geschlossen-Phase an, wobei die jeweilige offen-Phase und die darauf folgende geschlossen-Phase summiert werden. Dadurch wird klar, dass zu einer 19. Flanke keine solche Gesamtdauer angegeben werden kann, da die darauf folgende geschlossen-Phase beliebig lang sein kann. Spalte G weist auch hier auf die prozentuale Abweichung zum Sollwert hin. Der Bereich A3:D26 beschreibt die Prüfung des nsi/nsr-Kontakts, der Bereich von A29:D31 die Prüfung des nsa-Kontakts. In L10 wird zusätzlich angegeben, wie oft ein Kontaktprellen erkannt wurde. Die Werte für eine 0. Flanke stellen nur Startwerte dar, die nicht in die weitere Berechnung einfließen. Gleiches gilt für Werte einer 20. und folgenden Flanken, die einen Hinweis auf einen defekten nsa-Kontakt darstellen und bei einem funktionierenden Nummernschalter nicht auftreten dürfen. Aus diesem Grund ist es auch wichtig, dass eine 0 gewählt wird, weil z.B. einer gewählten 4 der einzige Hinweis auf einen defekten nsa-Kontakt der Hinweis „Gewählte Ziffer: 6“ in Zelle L10 ist.
Die Tabelle I9:O22 enthält Mittelwerte der gemessenen Daten, die jeweiligen Sollwerte sowie maximal zulässige Abweichung. Die Werte in den gelben Zellen können gem. den eigenen Anforderungen parametriert werden und werden bei der Auswertung der Ergebnisse im Bereich I28:O31 berücksichtigt. Die Fehlerbeseitigung bei einem defekten Nummernschalter erfolgt anhand der Tabellen A3:D26, A29:D31 sowie der Ergebnisauswertung in I28:O31, der Abgleich eines grundsätzlich funktionierenden Nummernschalters anhand der fett gedruckten Werte in Tabelle I9:O22 für den „Mittelwert offen“, „Mittelwert geschlossen“, „Dauer Wählvorgang“ und „Offen/geschlossen-Verhältnis“.

In die Zelle K2 kann eine Bezeichnung eingegeben werden, die den Nummernschalter bzw. das Telefon beschreibt. Dies macht einen Ausdruck aussagefähiger und dient zudem dem Wiederfinden gespeicherter Messwerte. Die Benamung des Feldes ist „KENNZ“.

Über das Feld L11 wird angegeben, wie oft von der Software ein Prellen des nsi-Kontakts bei der Messung erkannt wurde. Die Benamung des Feldes ist „PRELL“.

Über das Feld M7 wird ausgewählt, ob am Prüfgerät ein Nummernschalter (nsi/nsr- und nsa-Kontakt) oder ein Telefon angeschlossen ist. Im zweiten Fall werden nach der Messung die Spalten D und E ausgeblendet, da deren Werte wegen der verzerrenden Wirkung des Funkenlöschkondensators über keine Aussagekraft verfügen, und der nsa-Kontakt wird nicht in die Prüfung einbezogen. Es handelt sich um ein Feld mit vorgegebenen Parametern und Gültigkeitsprüfung. Die Benamung des Feldes ist „GERAET“.

In der Tabelle I28:O31 werden die Messwerte beurteilt, und es werden ggf. Reparaturhinweise gegeben. Basis für die entsprechenden Ausgaben ist das Tabellenblatt „Fehlerübersicht“. Die Benamung der beiden relevanten Felder I29 und I31 sind „STATUS“ und „COMMENT“.
Mit dem Button „Start“ wird ein Messzyklus gestartet. Innerhalb des im Tabellenblattes „Parameter“ unter „WARTEZEIT“ angegebenen Intervalls muss dann eine Ziffer gewählt werden, sonst bricht die Messung ab.
Durch Drücken des Buttons „Save“ werden die Messwerte unter Angabe der Kennzeichnung incl. Datum und Uhrzeit gespeichert und können so später wieder aufgerufen werden.

Das Rücksichern gespeicherter Werte erfolgt durch Drücken des Buttons „Load“.

Tabellenblatt „Speicherung“

Dieses Tabellenblatt dient der Speicherung von Messwerten, um diese zu einem späteren Zeitpunkt wieder in die Tabellen des Tabellenblattes „Test“ zu laden. Dadurch können Vergleiche desselben Nummernschalters zu unterschiedlichen Zeiten gezogen werden. Gespeichert werden der Name des gemessenen Geräts incl. Datum und Uhrzeit, die Werte aus den Tabellen A3:D26, A29:D31 sowie des Feldes L11 sowie der jeweilige Status der Messung im Feld „STATUS“ (I29) und der Reparaturhinweis im Feld „COMMENT“ (I31) des Tabellenblattes „Test“.
Tabellenblatt „Parameter“
Dieses Tabellenblatt enthält die grundlegenden Parameter der Software. Bei den gelben Feldern handelt es sich um Felder, in die frei Werte eingegeben werden können, bei dem hellgrünen Feld handelt es sich um ein Auswahlfeld.

Im Wert zur „PRELLZEIT“ wird die Zeitdauer in Sekunden angegeben, innerhalb derer aufeinanderfolgende Pegelwechsel des nsi/nsr-Kontaktes als Prellen gewertet werden (der nsa-Kontakt wird nicht auf Kontaktprellen hin untersucht).

Die „WARTEZEIT“ gibt an, wie viele Sekunden nach dem letzten Impuls das Ergebnis dargestellt wird bzw. wie lange man nach Drücken des „Start“ Buttons Zeit hat, einen Wählvorgang einzuleiten.

Mit „COMPORT“ (Auswahlfeld) wird bestimmt, über welchen seriellen Port die Messung erfolgt.

Da diese Einstellungen typischerweise beibehalten werden, empfiehlt es sich, nach Setzen der Parameter die Datei zu speichern. Das Tabellenblatt „Parameter“ braucht dann in der Regel nicht mehr zu beachtet werden.

Tabellenblatt „Versionsindex“
Hier werden die einzelnen Releases sowie Änderungen/Ergänzungen zwischen den Releaseversionen beschrieben.
Tabellenblatt „Fehlerübersicht“
Dieses Blatt stellt die Basis für das Testergebnis im Tabellenblatt „Test“, Tabelle I28:O31 dar. Hier werden die möglichen Fehler (Spalte D), die Möglichkeiten zur Erkennung anhand der Messwerte (Spalte E) sowie mögliche Abhilfemaßnahmen (Spalte G) angegeben. Spalte F beschreibt, ob im aktuellen Messzyklus die Fehlerbedingung erfüllt ist.

Jedem möglichen Fehler ist in Spalte A eine Fehlernummer zugeordnet. Wichtig ist in diesem Zusammenhang, dass die Fehlernummer umgekehrt zur Messreihenfolge steht. So ist beispielsweise die höchste Fehlernummer die 12, die eine fehlgeschlagene Messung beschreibt, denn was nützt es z.B. ein Impuls/Pausen-Verhältnis anzugeben, wenn gar kein Nummernschalter angeschlossen ist. Als nächstes werden die Startzustände von nsi/nsr- und nsa-Kontakt geprüft, denn was nützt es, wenn zwar der Nummernschalter ruhig läuft, jedoch der nsa-Kontakt von geschlossen nach geöffnet anstatt von geöffnet nach geschlossen wechselt. Wird die Tabelle um neue Fehler erweitert, so sind diese entsprechend dieser Vorgabe für Messreihenfolge/Fehlernummer zu ergänzen. Aufgrund des SVerweises in den Feldern D3:G3 sind die Fehler aufsteigend zu sortieren.
Die Formeln in den Spalten B und C bestimmen, ob der jeweilige Fehler bei dem jeweiligen Gerät (Nummernschalter bzw. Telefon) Berücksichtigung findet oder nicht. Dort, wo keine Formel eingegeben ist, wird der Fehler ignoriert, denn z.B. bei einem angeschlossenen Telefon kann der nsa-Kontakt nicht separat getestet werden.

Die höchste Fehlernummer mit dem jeweiligen Fehlertext sowie einer möglichen Abhilfe wird in Zeile 3 verdichtet. Aus dieser Zeile erhalten die Felder „STATUS“ und „COMMENT“ des Tabellenbalttes ihre Werte („STATUS“ aus D3, „COMMENT“ aus G3).

Das Feld „TESTNOTOK“, das angibt, ob überhaupt eine Messung möglich war, wird vom Programm gefüllt, die anderen Zellen in Spalte F über normale Excel-Funktionen.

Externe Erweiterungen
Hochgenaue Zeitmessung in Windows,
QueryPerformanceFrequency und QueryPerformanceCounter
Der Knowledge-Base Artikel

http://support.microsoft.com/kb/172338/de
beschreibt, wie man mittels Windows-eigener Mittel eine hochgenaue Zeitmessung vornehmen kann. Wenn Ihr System Zähler mit hoher Genauigkeit unterstützt (und bislang habe ich kein System gesehen, dass dies nicht tut), können Sie "QueryPerformanceCounter" und "QueryPerformanceFrequency" für diese Zeitmessungen verwenden.

QueryPerformanceCounter liefert die Anzahl der Einheiten des Zählers, die seit dem letzten Windows-Systemstart vergangen sind, QueryPerformanceFrequency gibt die Frequenz des hochgenauen Zählers an. Der Quotient von beiden besagt demnach, wie viele Sekunden seit dem letzten Windows-Systemstart vergangen sind.
Um die Funktionen verwenden zu können, ist es erforderlich, dass diese im Programm deklariert werden:

Declare Function QueryPerformanceFrequency Lib "kernel32" (lpFrequency As Currency) As Long
Declare Function QueryPerformanceCounter Lib "kernel32" (lpPerformanceCount As Currency) As Long

Unterstützt Ihr System keine hochgenaue Zeitmessung, so ist der Wert von QueryPerformanceFrequency gleich 0, und die Funktion TimeStamp bricht bei der erfolglosen Division mit einem Fehler ab.
Schnittstelle Excel <-> RS 232: RSAPI.DLL
Um mittels Excel auf die serielle Schnittstelle zugreifen zu können, bedarf es zusätzlicher Software wie z.B. RSAPI.DLL oder MSCOMM. Für die Nummernschalter-Prüfsoftware wurde die RSAPI.DLL verwendet, die von H.-J. Berndt und B. Kainka für das Buch „Messen, Steuern, Regeln mit Word und Excel“ entwickelt wurde.
Stand Mai 2009 konnte die Datei unter

http://www.b-kainka.de/download.htm und http://www.b-kainka.de/rsupdate.htm (Update)

für den privaten Einsatz und den Einsatz in Schulen kostenlos downgeloaded werden. Das Zip-File enthält eine Hilfedatei, in der die Funktionalitäten der DLL beschrieben werden.
Für die Nummernschalter-Prüfsoftware werden folgende Funktionalitäten verwendet:

OPENCOM
Öffnet eine RS232-Schnittstelle mit dem übergebenen Parameterstring. Die Schnittstelle wird vorher geschlossen, RTS wird abgeschaltet. Die angesprochene Schnittstelle wird in der Variablen strCOMPort übergeben. Ist z.B. strCOMPort=COM1, dann wird mit OPENCOM(strCOMPort & ":38400,N,8,2") COM-Port 1 mit einer Geschwindigkeit von 38.400 Bits/s, keine Parität, 8 Datenbits und 2 Stoppbits geöffnet.
RTS 1
Schaltet die Leitung RTS für die mit OPENCOM geöffnete Schnittstelle auf logisch 1. Hiermit wird der mit CTS zu messende Pegel zur Verfügung gestellt.
CTS
Liest die Leitung CTS der mit OPENCOM geöffneten Schnittstelle.

DSR
Liest die Leitung DSR der mit OPENCOM geöffneten Schnittstelle.

CLOSECOM
Schließt die Schnittstelle. Danach können andere Geräte die Schnittstelle benutzen.

Um die Funktionen verwenden zu können, ist es erforderlich, dass diese im Programm zuvor deklariert werden:

Declare Function OPENCOM Lib "RSAPI.DLL" (ByVal Parameter$) As Integer
Declare Function CLOSECOM Lib "RSAPI.DLL" () As Integer
Declare Function CTS Lib "RSAPI.DLL" () As Integer
Declare Function DSR Lib "RSAPI.DLL" () As Integer
Declare Sub RTS Lib "RSAPI.DLL" (ByVal EinsNull As Integer)
Erläuterungen zum Quellcode
Der Quellcode verteilt sich über unterschiedliche Bereiche des Projekts:

· Im Modul1 befindet sich der Code für das Hauptprogramm.
· Im Excel-Objekt Tabelle „Test“ befindet sich der Code zu den Buttons „Start“, „Save“ und „Load“.
· Im Formular „UserForm1“ befindet sich der Code für das Fenster, mit dem die Daten für eine Rücksicherung ausgewählt werden.
Hauptprogramm

Das Hauptprogramm befindet sich in Modul1. Zunächst werden die Funktionen für die Verwendung der RSAPI.DLL sowie des hochgenauen Zeitmessers deklariert, damit sie im folgenden aufgerufen werden können:

Declare Function OPENCOM Lib "RSAPI.DLL" (ByVal Parameter$) As Integer
Declare Function CLOSECOM Lib "RSAPI.DLL" () As Integer
Declare Function CTS Lib "RSAPI.DLL" () As Integer
Declare Function DSR Lib "RSAPI.DLL" () As Integer
Declare Sub RTS Lib "RSAPI.DLL" (ByVal EinsNull As Integer)
Declare Function QueryPerformanceFrequency Lib "kernel32" (lpFrequency As Currency) As Long
Declare Function QueryPerformanceCounter Lib "kernel32" (lpPerformanceCount As Currency) As Long

Die Funktion „TimeStamp” dient dazu, die vom nsi/nsr-Kontakt erzeugten Flankenwechsel mit einem Zeitstempel zu versehen. Die Dauer eines Impulses wird dabei als Differenz zwischen dem aktuellen und dem vorangegangenen Zeitstempel bestimmt.
Aufgrund der Genauigkeitsanforderung sind der Typ der Funktion sowie der in ihr verwendeten Variablen als Currency definiert.

QueryPerformanceFrequency currFreq
QueryPerformanceCounter currTimer
TimeStamp = currTimer / currFreq

Unterstützt Ihr System keine hochgenaue Zeitmessung, so ist der Wert von QueryPerformanceFrequency gleich 0, und die Funktion TimeStamp bricht bei der erfolglosen Division mit einem Fehler ab.

Es werden die Frequenz des hochgenauen Zählers sowie die aktuelle Anzahl der seit dem letzten Windows-Start vergangenen Impulse ermittelt. Der Quotient der beiden ist die Anzahl der seit dem letzten Windows-Start vergangenen Sekunden, dieser wird von der Funktion als Zeitstempel zurückgegeben.

Zu Beginn des Hauptprogramms „Test“ werden die Variablen deklariert:
Dim arrnsa(3, 2) As Double
Dim arrnsinsr(24, 2) As Double

Hierbei handelt es sich um die Eingabefelder betreffend den nsa und den nsi/nsr-Kontakt, in die die gemessenen Daten zwischengespeichert werden. Die Felder werden vor allem aus Performancegründen verwendet und um die Messung nicht durch zusätzliche API-Aufrufe zu verfälschen.
In die Feldkomponenten (x,1) wird der Pegel der CTS-/DSR-Leitung, in die Feldkomponenten (x,2) der Zeitstempel geschrieben. Am Ende der Messung erfolgt ein Übertrag der Feldwerte in die Tabellen des Blattes „Test“.

Dim intCTSSpeicher As Integer
Dim intDSRSpeicher As Integer

Im Rahmen der Messung wird ständig das Signal auf der CTS- bzw. DSR-Leitung daraufhin untersucht, ob ein Flankenwechsel stattgefunden hat. Dazu ist es erforderlich, die alten Werte zu speichern.

Dim intCTSneu As Integer
Dim intDSRneu As Integer

Diese Variablen dienen zum Speichern des aktuellen CTS- bzw. DSR-Wertes, um diese mit dem alten Wert zu vergleichen.

Dim intznsa As Integer
Dim intznsinsr As Integer

Mittels dieser Zähler werden Feldoperationen in den Feldern arrnsa und arrnsinsr durchgeführt.

Dim intLaufvar1 As Integer
Dim intLaufvar2 As Integer
Zwei Integer-Variablen zur Verwendung als universelle Laufvariablen.

Dim intPrellCounter As Integer

Diese Variable beinhaltet die Anzahl der 0/1- bzw. 1/0-Wechsel des nsi/nsr-Kontakts (CTS-Leitung), die als Prellen identifiziert wurden. Zur Ausgabe im Tabellenblatt „Test“ wird der betreffende Wert halbiert.

Dim dPrellzeit As Double
Dim dWartezeit As Double

In diesen Variablen werden die Zeitdauer, innerhalb derer Kontaktwechsel als Prellen gewertet werden, sowie die Wartezeit bis zum Abschluss des Tests gespeichert. Die Werte werden dem Tabellenblatt „Parameter“ in den Zellen „PRELLZEIT“ und „WARTEZEIT“ entnommen.
Dim bTestOK As Boolean
Diese Variable gibt an, ob eine erfolgreiche Messung durchgeführt wurde.
Sheets("Test").Range("Status") = "Messung läuft ..."
Sheets("Test").Range("Comment") = ""
Application.Calculation = xlManual
Application.ScreenUpdating = False

Zunächst wird das ehemalige Messergebni in der Ausgabe gelöscht. Aus Performancegründen und um die Messungen nicht zu beeinflussen werden die automatische Bildschirmaktualisierung sowie die automatische Berechnung ausgeschaltet.

Sheets("Test").Range("A3:D26").ClearContents
Sheets("Test").Range("A29:D31").ClearContents
Sheets("Test").Range("PRELL").ClearContents
If Sheets("Test").Range("GERAET") = "Telefon" Then
    Sheets("Test").Columns("D:E").Hidden = True
Else
    Sheets("Test").Columns("D:E").Hidden = False
End If
For intLaufvar1 = 0 To 1
    For intLaufvar2 = 0 To 23
        arrnsinsr(intLaufvar2, intLaufvar1) = 0
    Next intLaufvar2
    For intLaufvar2 = 0 To 2
        arrnsinsr(intLaufvar2, intLaufvar1) = 0
    Next intLaufvar2
Next intLaufvar1
intPrellCounter = 0
Sheets("Fehlerübersicht").Range("TESTNOTOK").Value = False
dPrellzeit = Sheets("Parameter").Range("PRELLZEIT")
dWartezeit = Sheets("Parameter").Range("WARTEZEIT")

Zunächst werden die Ausgabefelder im Tabellenblatt „Test“ einer evt. alten Messung gelöscht. Handelt es sich bei dem angeschlossenen Gerät um ein Telefon (und nicht um einen Nummernschalter), werden die Spalten D und E des Tabellenblatts „Test“ ausgeblendet, das sie aufgrund der verzerrenden Wirkung des Funkenlöschkondensators keine Aussagekraft besitzen. Anschließend werden die Eingabefelder für den nsi/nsr- und den nsa-Kontakt gelöscht. Der Zähler für das Kontaktprellen wird vorbesetzt, die Werte für „PRELLZEIT“ und „WARTEZEIT“ werden aus dem Tabellenblatt „Parameter“ übernommen.
OPENCOM (Sheets(1).Range("COMPORT") & ":38400,N,8,2")
RTS 1

Anschließend wird die serielle Schnittstelle parametriert und geöffnet. Die RTS-Leitung der seriellen Schnittstelle wird mit einem Pegel beaufschlagt, der im Folgenden über die CTS- und die DSR-Leitung gemessen werden kann.

intCTSSpeicher = CTS
arrnsinsr(0, 0) = intCTSSpeicher
arrnsinsr(0, 1) = TimeStamp
intDSRSpeicher = DSR
arrnsa(0, 0) = intDSRSpeicher
arrnsa(0, 1) = TimeStamp

bTestOK = False
intznsinsr = 0
intznsa = 0

Die Zwischenspeicher zur Speicherung des Wertes der CTS- und der DSR-Leitung werden initialisiert. In die Eingabefelder werden der logische Zustand dieser Leitungen sowie die Zeitstempel zu Beginn der Messung (nach Drücken des Buttons „Start“) gespeichert, die Abbruchbedingung für die Testroutine sowie die Zähler für Operationen im Eingabefeld werden initialisiert.

In der nachfolgenden While-Schleife werden permanent die CTS- und die DSR-Leitung auf einen Flankenwechsel hin getestet, bis die mit „WARTEZEIT“ angegebene Zeitdauer kein Flankenwechsel stattgefunden hat.

While bTestOK = False
    intCTSneu = CTS
    intDSRneu = DSR
    If intCTSneu <> intCTSSpeicher Then
        intCTSSpeicher = intCTSneu
        If TimeStamp - arrnsinsr(intznsinsr, 1) < dPrellzeit Then
            intPrellCounter = intPrellCounter + 1
        Else
            arrnsinsr(intznsinsr + 1, 0) = intCTSSpeicher
            arrnsinsr(intznsinsr + 1, 1) = TimeStamp
            intznsinsr = intznsinsr + 1
            If intznsinsr > 23 Then
                bTestOK = True
            End If
        End If
    End If
    If Sheets("Test").Range("GERAET") = "Nummernschalter" Then
        If intDSRneu <> intDSRSpeicher Then
            intDSRSpeicher = intDSRneu
            If TimeStamp - arrnsa(intznsa, 1) > dPrellzeit Then
                arrnsa(intznsa + 1, 0) = intDSRSpeicher
                arrnsa(intznsa + 1, 1) = TimeStamp
                intznsa = intznsa + 1
            End If
        End If
    End if
    If TimeStamp - arrnsinsr(intznsinsr, 1) > dWartezeit Then
        If intznsinsr <> 0 Then
            bTestOK = True
        Else
            Sheets("Fehlerübersicht").Range("TESTNOTOK").Value = True
            GoTo Exit_Test
        End If
    End If
Wend

Zunächst werden der aktuelle Zustand der CTS- und der DSR-Leitung eingelesen. Anschließend wird jeweils für beiden Leitungen überprüft, ob ein Flankenwechsel stattgefunden hat. War dies der Fall, so wird der neue Zustand in den Vergleichsspeicher intCTSSpeicher bzw. intDSRSpeicher geladen. Beim Laden dieser Speicher müssen zwingend die Variable intCTSneu bzw. intDSRneu und dürfen nicht die Aufrufe CTS bzw. DTS verwendet werden, da es sein kann, dass die Leitungen zwischen der if-Abfrage und dem anschließenden Speichern im Vergleichsspeicher ihren Pegel gewechselt haben.
Ist der Zeitstempel des vorangegangenen Flankenwechsels kürzer als der über „PRELLZEIT“ definierte Wert, so wird von einem Prellen des Kontakts ausgegangen. Im Falls des nsi/nsr-Kontakts wird dieses Prellen gezählt und unterdrückt, im Fall des nsa-Kontaktes wird es nur unterdrückt. Der neue Zustand der Leitungen incl. Zeitstempel wird in die Eingabefelder geschrieben, und die Indices zum Zugriff auf die Eingabefelder werden inkrementiert.
Treten auf der CTS-Leitung (nsi/nsr-Kontakt) mehr als 23 Impulse auf, so wird die Messung abgebrochen, da ansonsten die Feldgrenzen des betreffenden Eingabefeldes überschritten werden.

Wurde in der durch „WARTEZEIT“ angegebenen Zeitdauer kein Impuls detektiert, so wird die Messung beendet. Wurde überhaupt kein Impuls erkannt, wird die Messung zudem als erfolglos bewertet. Die Prozedur wird dann über die Sprungmarke „Exit_Test“ verlassen.

Sheets("Test").Range("A3").Select   'Ausgabefeld nsi/nsr
ActiveCell.Value = 0
ActiveCell.Offset(0, 1).Value = arrnsinsr(0, 1)
ActiveCell.Offset(0, 2).Value = arrnsinsr(0, 0)
For intLaufvar1 = 1 To IIf(intznsinsr > 23, 23, intznsinsr - 1)
    ActiveCell.Offset(intLaufvar1, 0).Value = intLaufvar1
    ActiveCell.Offset(intLaufvar1, 1).Value = arrnsinsr(intLaufvar1, 1)
    ActiveCell.Offset(intLaufvar1, 2).Value = arrnsinsr(intLaufvar1, 0)
    ActiveCell.Offset(intLaufvar1, 3).Value = (arrnsinsr(intLaufvar1 + 1, 1) - arrnsinsr(intLaufvar1, 1)) * 1000
Next intLaufvar1
If Sheets("Test").Range("GERAET") = "Nummernschalter" Then
    Sheets("Test").Range("A29").Select  'Ausgabefeld nsa
    ActiveCell.Value = 0
    ActiveCell.Offset(0, 1).Value = arrnsa(0, 1)
    ActiveCell.Offset(0, 2).Value = arrnsa(0, 0)
    ActiveCell.Offset(1, 0).Value = 1
    ActiveCell.Offset(1, 1).Value = arrnsa(1, 1)
    ActiveCell.Offset(1, 2).Value = arrnsa(1, 0)
    ActiveCell.Offset(1, 3).Value = (arrnsa(2, 1) - arrnsa(1, 1)) * 1000
    ActiveCell.Offset(2, 0).Value = 2
    ActiveCell.Offset(2, 1).Value = arrnsa(2, 1)
    ActiveCell.Offset(2, 2).Value = arrnsa(2, 0)
End If
Sheets("Test").Range("PRELL").Value = intPrellCounter / 2
Sheets("Test").Range("KENNZ").Select

Im nachfolgenden Programmteil werden die Ausgabefelder im Tabellenblatt „Test“ befüllt.
CLOSECOM
Application.ScreenUpdating = True
Application.Calculation = xlCalculationAutomatic
Sheets("Test").Range("STATUS") = Sheets("Fehlerübersicht").Range("D3")
Sheets("Test").Range("COMMENT").Value = Sheets("Fehlerübersicht").Range("G3").Value
Exit Sub

Der Befehl „CLOSECOM“ schließt die verwendete seriellen Schnittstelle und gibt diese für andere Anwendungen frei. Im Anschluss an die Messung werden die Bildschirmaktualisierung sowie die automatische Berechnung wieder eingeschaltet sowie die Gesamtbewertung der Messung in das Tabellenblatt „Test“ übernommen.
If intznsa = 3 Then
    MsgBox ("Zu häufige Impulse nsa-Kontakt. Evt. Anschlüsse nsi/nsr- und nsa-Kontakt vertauscht.")
Else
    MsgBox Err.Description, vbExclamation
    MsgBox ("NSI/NSR-COUNTER: " & intznsinsr & ", NSA-COUNTER: " & intznsa)
End If
Resume Exit_Test

Läuft die Prozedur auf einen Fehler, werden entsprechende Hinweise ausgegeben, die eine Lokalisierung des Fehlers gestatten.

Programmteile im Excel-Objekt Tabelle „Test“

Hier findet sich der Code zu den Buttons „Start“, „Save“ und „Load“.

Button Start:

Call Modul1.Test

Über diesen Button wird lediglich das Hauptprogramm aufgerufen.

Button Save:
Dieser Button dient dazu, ermittelte Messwerte im Tabellenblatt „Speicherung“ abzulegen.

Zunächst wird im Tabellenblatt „Speicherung“ die Bezeichnung des getesteten Geräts mit Datum und Uhrzeit gespeichert, indem dies in Spalte A an die der bislang letzten Zeile folgenden Zeile geschrieben wird:

Sheets("Speicherung").Select
Sheets("Speicherung").Range("A" & CStr(ActiveSheet.UsedRange.Rows.Count + 1)).Value = Sheets("Test").Range("KENNZ").Value & ", " & Date & ", " & Time
Anschließend erfolgt das Speichern der Flanken, Zeitstempel, Tegel und Messwerte, und zwar getrennt für den nsi/nsr- und den nsa-Kontakt. Dazu werden die entsprechenden Bereiche im Tabellenblatt „Test“ kopiert und dann transponiert im Tabellenblatt „Speicherung“ abgelegt:
'Flanken nsi/nsr speichern
Sheets("Test").Select
ActiveSheet.Range("A3:A26").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("B" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Zeitstempel nsi/nsr speichern
Sheets("Test").Select
ActiveSheet.Range("B3:B26").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("Z" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Pegel nsi/nsr speichern
Sheets("Test").Select
ActiveSheet.Range("C3:C26").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("AX" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Messwerte nsi/nsr speichern
Sheets("Test").Select
ActiveSheet.Range("D3:D26").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("BV" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

'Flanken nsa speichern
Sheets("Test").Select
ActiveSheet.Range("A29:A31").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("CT" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Zeitstempel nsa speichern
Sheets("Test").Select
ActiveSheet.Range("B29:B31").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("CW" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Pegel nsa speichern
Sheets("Test").Select
ActiveSheet.Range("C29:C31").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("CZ" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True
'Messwerte nsa speichern
Sheets("Test").Select
ActiveSheet.Range("D29:D31").Select
Selection.Copy
Sheets("Speicherung").Select
ActiveSheet.Range("DC" & CStr(ActiveSheet.UsedRange.Rows.Count)).Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

Zuletzt werden das Gerät, der Status und der Reparaturhinweis gespeichert

'Gerät speichern
ActiveSheet.Range("DF" & CStr(ActiveSheet.UsedRange.Rows.Count)).Value = Sheets("Test").Range("GERAET").Value
'Anzahl Prellen speichern
ActiveSheet.Range("DG" & CStr(ActiveSheet.UsedRange.Rows.Count)).Value = Sheets("Test").Range("PRELL").Value
'Status und Comment speichern
Sheets("Speicherung").Range("DH" & CStr(ActiveSheet.UsedRange.Rows.Count)).Value = Sheets("Test").Range("STATUS").Value
Sheets("Speicherung").Range("DI" & CStr(ActiveSheet.UsedRange.Rows.Count)).Value = Sheets("Test").Range("COMMENT").Value

und das Feld „KENNZ” im Tabellenblatt „Test” ausgewählt.

Application.CutCopyMode = False
Sheets("Test").Select
ActiveSheet.Range("KENNZ").Select
Button Load:
Dieser Button löst das Rücklesen gespeicherter Messwerte in das Tabellenblatt „Test” aus.

UserForm1.Show
Dazu wird das „UserForm1“ gestartet.

Programmteile im „UserForm1“

Nach Aufruf des UserForms wird zunächst die Auswahl-Box zur Auswahl der gespeicherten Messungen befüllt. Da dies mit Wechseln des Tabellenblatts verbunden ist, wird die automatische Bildschirmaktualisierung zunächst ausgeschaltet.
Application.ScreenUpdating = False

Anschließend erhält die Auswahl-Box ihre Werte, indem diese sukzessiv aus Spalte A des Tabellenblattes „Speicherung” eingelesen werden.

Sheets(“Speicherung”).Select
ActiveSheet.Range("A2").Select
While ActiveCell.Value <> ""
    ComboBox1.AddItem ActiveCell.Value
    ActiveCell.Offset(1, 0).Select
Wend

Zum Abschluss werden die automatische Bildschirmaktualisierung wieder eingeschaltet und das Feld „KENNZ” im Tabellenblatt „Test” ausgewählt.
Application.ScreenUpdating = True
Sheets(“Test”).Select
ActiveSheet.Range("KENNZ").Select

Nach Initialisierung der Auswahl-Box erfolgt nach einer Auswahl der Messung auf das Ereignis „ComboBox1_Change“ hin der eigentliche Rückladevorgang. Auch hierzu wird sinnvoller Weise die automatische Bildschirmaktualisierung zunächst ausgeschaltet. Anschließend wird ermittelt, in welcher Zeile des Tabellenblatts „Speicherung“ die gewünschten Messwerte stehen, damit auf diese zugegriffen werden kann. Dazu wird die Spalte A dieses Tabellenblattes so lange gesucht, bis der Inhalt der Auswahl-Box gefunden ist. Der Zeilenindex wird in der Variablen „intZeile“ gespeichert.
Sheets(“Speicherung”).Select
ActiveSheet.Range("A2").Select
While ActiveCell.Value <> ComboBox1.Value
    ActiveCell.Offset(1, 0).Select
Wend
intZeile = ActiveCell.Row

Analog zum Speichern der Messwerte werden nun die Bezeichung des getesteten Geräts sowie alle Messwerte (Flanken, Zeitstempel, Pegel, Impulsdauern, Anzahl der Prellungen, Status und Reparaturhinweis) ins Tabellenblatt „Test“ übertragen, die automatische Bildschirmaktualisierung wird wieder eingeschaltet, und das Feld „KENNZ” im Tabellenblatt „Test” wird ausgewählt.
'Flanken nsi/nsr zurückladen
Sheets("Speicherung").Select
ActiveSheet.Range("B" & CStr(intZeile) & ":Y" & CStr(intZeile)).Select
Selection.Copy
Sheets("Test").Select
ActiveSheet.Range("A3:A26").Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

'Zeitstempel nsi/nsr zurückladen
Sheets("Speicherung").Select
ActiveSheet.Range("Z" & CStr(intZeile) & ":AW" & CStr(intZeile)).Select
Selection.Copy
Sheets("Test").Select
ActiveSheet.Range("B3:B26").Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

'Logische Pegel nsi/nsr zurückladen
Sheets("Speicherung").Select
ActiveSheet.Range("AX" & CStr(intZeile) & ":BU" & CStr(intZeile)).Select
Selection.Copy
Sheets("Test").Select
ActiveSheet.Range("C3:C26").Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

'Zeitdauern nsi/nsr zurückladen
Sheets("Speicherung").Select
ActiveSheet.Range("BV" & CStr(intZeile) & ":CS" & CStr(intZeile)).Select
Selection.Copy
Sheets("Test").Select
ActiveSheet.Range("D3:D26").Select
Selection.PasteSpecial Paste:=xlPasteValues, Operation:=xlNone, SkipBlanks:=False, Transpose:=True

'Signale nsa zurückladen
ActiveSheet.Range("A29").Value = Sheets("Speicherung").Range("CT" & CStr(intZeile)).Value
ActiveSheet.Range("A30").Value = Sheets("Speicherung").Range("CU" & CStr(intZeile)).Value
ActiveSheet.Range("A31").Value = Sheets("Speicherung").Range("CV" & CStr(intZeile)).Value
ActiveSheet.Range("B29").Value = Sheets("Speicherung").Range("CW" & CStr(intZeile)).Value
ActiveSheet.Range("B30").Value = Sheets("Speicherung").Range("CX" & CStr(intZeile)).Value
ActiveSheet.Range("B31").Value = Sheets("Speicherung").Range("CY" & CStr(intZeile)).Value
ActiveSheet.Range("C29").Value = Sheets("Speicherung").Range("CZ" & CStr(intZeile)).Value
ActiveSheet.Range("C30").Value = Sheets("Speicherung").Range("DA" & CStr(intZeile)).Value
ActiveSheet.Range("C31").Value = Sheets("Speicherung").Range("DB" & CStr(intZeile)).Value
ActiveSheet.Range("D30").Value = Sheets("Speicherung").Range("DD" & CStr(intZeile)).Value

'Gerät, Anzahl Prellen, Status und Reparaturhinweis zurückladen
ActiveSheet.Range("GERAET").Value = Sheets("Speicherung").Range("DF" & CStr(intZeile)).Value
ActiveSheet.Range("PRELL").Value = Sheets("Speicherung").Range("DG" & CStr(intZeile)).Value
ActiveSheet.Range("STATUS").Value = Sheets("Speicherung").Range("DH" & CStr(intZeile)).Value
ActiveSheet.Range("COMMENT").Value = Sheets("Speicherung").Range("DI" & CStr(intZeile)).Value

Application.CutCopyMode = False
Application.ScreenUpdating = True
Sheets("Test").Select
ActiveSheet.Range("KENNZ").Select

